

Gramática BNF reconhecida pelo **compilador**.

<codigo> ::= "PROGRAMA" <espaco_em_branco> <nome_programa> ":"
<quebra_linha> <corpo_codigo>

<nome_programa> ::= "\"" <nome_simples> "\""

<nome_simples> ::= <letra_inicial> <nome_extendido_opt>
<nome_extendido_opt> ::= <caractere_valido> *

<corpo_codigo> ::= "INICIO" <quebra_linha> <linhas_de_codigo> <linha_res> "FIM"
<espaco_em_branco_opcional> <quebra_linha_opcional>

<linhas_de_codigo> ::= (<linha_variavel>)*

<linha_variavel> ::= <atribuicao_variavel> <espaco_em_branco_opcional>
<quebra_linha>

<atribuicao_variavel> ::= <nome_simples> <espaco_em_branco_opcional> "="
<espaco_em_branco_opcional> <expressao_aritmetica>

<linha_res> ::= <atribuicao_resultado> <espaco_em_branco_opcional>
<quebra_linha>

<atribuicao_resultado> ::= "RES" <espaco_em_branco_opcional> "="
<espaco_em_branco_opcional> <expressao_aritmetica>

<expressao_aritmetica> ::= <bloco_termos> (<espaco_em_branco_opcional> ("+" | "-")
<espaco_em_branco_opcional> <bloco_termos>) *

<bloco_termos> ::= <elemento> (<espaco_em_branco_opcional> ("*" | "/")
<espaco_em_branco_opcional> <elemento>) *

<elemento> ::= "(" <espaco_em_branco_opcional> <expressao_aritmetica>
<espaco_em_branco_opcional> ")" | <digito> + | <nome_simples>

<letra_inicial> ::= [A-Z] | [a-z]

<digito> ::= [0-9]

<caractere_valido> ::= <letra_inicial> | <digito> | "_"

<espaco_em_branco> ::= (" " | "\t")

<espaco_em_branco_opcional> ::= <espaco_em_branco> *

<quebra_linha> ::= "\n"

<quebra_linha_opcional> ::= <quebra_linha> ?